Example 8.14 Heat Conduction in a Cylinder
Equation (8.1.30) is solved in Maple below:

```
> restart : with(inttrans) : with(plots) :
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2)+1/x*diff(u(x,t),x);
```

$$eq := \frac{\partial}{\partial t} u(x,t) = \frac{\partial^2}{\partial x^2} u(x,t) + \frac{\frac{\partial}{\partial x} u(x,t)}{x} \tag{1}$$

```
> u(x,0):=0;
```

$$u(x,0) := 0 \tag{2}$$

```
> bc1:=diff(u(x,t),x)=0;
```

$$bc1 := \frac{\partial}{\partial x} u(x,t) = 0 \tag{3}$$

```
> bc2:=u(x,t)=1;
```

$$bc2 := u(x,t) = 1 \tag{4}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\,U(x) = \frac{d^2}{dx^2} U(x) + \frac{\frac{d}{dx} U(x)}{x} \tag{5}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := \frac{d}{dx} U(x) = 0 \tag{6}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := U(x) = \frac{1}{s} \tag{7}$$

```
> dsolve(eqs,U(x));
```

$$U(x) = \_C1\, \mathrm{BesselJ}\left(0, \sqrt{-s}\, x\right) + \_C2\, \mathrm{BesselY}\left(0, \sqrt{-s}\, x\right) \tag{8}$$

Since BesselY becomes infinite x = 0_C2 should be zero and the solution is taken as:

```
> U(x):=c[1]*BesselJ(0,(-s)^(1/2)*x);
```

$$U(x) := c_1\, \mathrm{BesselJ}\left(0, \sqrt{-s}\, x\right) \tag{9}$$

```
> eq0:=eval(subs(x=0,bc1));
```

$$eq0 := 0 = 0 \tag{10}$$

```
> eq1:=eval(subs(x=1,bc2));
```

$$eq1 := c_1\, \mathrm{BesselJ}\left(0, \sqrt{-s}\right) = \frac{1}{s} \tag{11}$$

```
> con:=solve({eq1},{c[1]}):
```

```
> U(x):=subs(con,U(x)):
> U(x):=factor(simplify(U(x)));
```

$$U(x) := \frac{\mathrm{BesselJ}\left(0, \sqrt{-s}\, x\right)}{\mathrm{BesselJ}\left(0, \sqrt{-s}\right) s} \quad (12)$$

```
> P(s):=numer(U(x));
```

$$P(s) := \mathrm{BesselJ}\left(0, \sqrt{-s}\, x\right) \quad (13)$$

```
> Q(s):=denom(U(x));
```

$$Q(s) := \mathrm{BesselJ}\left(0, \sqrt{-s}\right) s \quad (14)$$

```
> solve(Q(s),s);
```

$$RootOf\left(\mathrm{BesselJ}\left(0, \sqrt{-\_Z}\right)\right), 0 \quad (15)$$

```
> eig:=BesselJ(0,(-s)^(1/2));
```

$$eig := \mathrm{BesselJ}\left(0, \sqrt{-s}\right) \quad (16)$$

```
> convert(eig,BesselI);
```

$$\mathrm{BesselJ}\left(0, \sqrt{-s}\right) \quad (17)$$

```
> eiglambda:=simplify(subs(s^(1/2)=I*lambda,(-s)^(1/2)=lambda,s=-
  lambda^2,eig));
```

$$eiglambda := \mathrm{BesselJ}(0, \lambda) \quad (18)$$

```
> plot(eiglambda,lambda=0..20,thickness=3,axes=boxed):
```

The roots are:

```
> 0,-lambda^2;
```

$$0, -\lambda^2 \quad (19)$$

```
> N:=20;
```

$$N := 20 \quad (20)$$

```
> l[1]:=fsolve(eiglambda,lambda=0..3);
```

$$l_1 := 2.404825558 \quad (21)$$

```
> for i from 2 to N do l[i]:=fsolve(eiglambda,lambda=l[i-1]..l
  [i-1]+4);od:
> seq(l[i],i=1..N);
```

2.404825558, 5.520078110, 8.653727913, 11.79153444, 14.93091771, 18.07106397, 21.21163663, 24.35247153, 27.49347913, 30.63460647, 33.77582021, 36.91709835, 40.05842576, 43.19979171, 46.34118837, 49.48260990, 52.62405184, 55.76551076, 58.90698393, 62.04846919 **(22)**

```
> A(s):=P(s)/diff(Q(s),s):
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := \frac{2\,\mathrm{BesselJ}\left(0, \sqrt{-\mu}\,x\right)\sqrt{-\mu}}{\mathrm{BesselJ}\left(1, \sqrt{-\mu}\right)\mu + 2\,\mathrm{BesselJ}\left(0, \sqrt{-\mu}\right)\sqrt{-\mu}} \tag{23}$$

```
> A[0]:=limit(A[n],mu=0);
```

$$A_0 := 1 \tag{24}$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*lambda,(-mu)^(1/2)=lambda,mu^
(3/2)=-I*lambda^3,mu=-lambda^2,A[n]));
```

$$A_n := -\frac{2\,\mathrm{BesselJ}(0, \lambda x)}{\mathrm{BesselJ}(1, \lambda)\,\lambda - 2\,\mathrm{BesselJ}(0, \lambda)} \tag{25}$$

```
> u0s:=A[0]*subs(mu=0,1/(s-mu));
```

$$u0s := \frac{1}{s} \tag{26}$$

```
> u0t:=invlaplace(u0s,s,t);
```

$$u0t := 1 \tag{27}$$

```
> uns:=A[n]/(s-mu);
```

$$uns := -\frac{2\,\mathrm{BesselJ}(0, \lambda x)}{\left(\mathrm{BesselJ}(1, \lambda)\,\lambda - 2\,\mathrm{BesselJ}(0, \lambda)\right)(s - \mu)} \tag{28}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := \frac{2\,\mathrm{BesselJ}(0, \lambda x)\,\mathrm{e}^{\mu t}}{-\mathrm{BesselJ}(1, \lambda)\,\lambda + 2\,\mathrm{BesselJ}(0, \lambda)} \tag{29}$$

```
> unt:=subs(mu=-l[n]^2,lambda=l[n],unt);
```

$$unt := \frac{2\,\mathrm{BesselJ}(0, l_n x)\,\mathrm{e}^{-l_n^2 t}}{-\mathrm{BesselJ}(1, l_n)\,l_n + 2\,\mathrm{BesselJ}(0, l_n)} \tag{30}$$

The following solution and plots are obtained:

```
> U:=u0t+Sum(unt,n=1..infinity);
```

$$U := 1 + \sum_{n=1}^{\infty} \frac{2\,\mathrm{BesselJ}(0, l_n x)\,\mathrm{e}^{-l_n^2 t}}{-\mathrm{BesselJ}(1, l_n)\,l_n + 2\,\mathrm{BesselJ}(0, l_n)} \tag{31}$$

```
> u:=piecewise(t=0,0,t>0,subs(infinity=20,U)):
> u:=evalf(u):
> plot3d(u,x=0..1,t=0..0.5,axes=boxed,title="Figure Exp. 8.26.",
labels=[x,t,"u"],orientation=[-145,60]);
```

Figure Exp. 8.26.




```
> plot([subs(t=0,u),subs(t=0.1,u),subs(t=0.2,u),subs(t=0.5,u)],x=
  0..1,title="Figure Exp. 8.27.",axes=boxed,thickness=5,labels=[x,
  "u"]);
```

Figure Exp. 8.27.




```
>
```